# $Id$
# Maintainer: Allan McRae <allan@archlinux.org>

pkgname=isl
pkgver=0.13
pkgrel=1
pkgdesc="Library for manipulating sets and relations of integer points bounded by linear constraints"
arch=('i686' 'x86_64')
url="http://freecode.com/projects/isl"
depends=('gmp')
license=('MIT')
source=(http://isl.gforge.inria.fr/$pkgname-$pkgver.tar.bz2)
md5sums=('e4cf20c4c10d5d613457558b1f46b5e2')


build() {
  cd "$srcdir/$pkgname-$pkgver"
  ./configure --prefix=/usr
  make
}

check() {
  cd "$srcdir/$pkgname-$pkgver"
  make check
}

package() {
  cd "$srcdir/$pkgname-$pkgver"
  make DESTDIR="$pkgdir/" install

  install -dm755 $pkgdir/usr/share/gdb/auto-load/usr/lib/
  mv $pkgdir/usr/lib/libisl.so.*-gdb.py $pkgdir/usr/share/gdb/auto-load/usr/lib/
  
  install -Dm644 LICENSE $pkgdir/usr/share/licenses/isl/LICENSE
}
